Using Maple's Groebner Basis Package.

```
> with(Groebner);
```

[*Basis*, *FGLM*, *HilbertDimension*, *HilbertPolynomial*, *HilbertSeries*, *Homogenize*, *InitialForm*, *InterReduce*, *IsBasis*, *IsProper*, *IsZeroDimensional*, *LeadingCoefficient*, *LeadingMonomial*, *LeadingTerm*, *MatrixOrder*, *MaximalIndependentSet*, *MonomialOrder*, *MultiplicationMatrix*, *MultivariateCyclicVector*, *NormalForm*, *NormalSet*, *RationalUnivariateRepresentation*, *Reduce*, *RememberBasis*, *SPolynomial*, *Solve*, *SuggestVariableOrder*, *Support*, *TestOrder*, *ToricIdealBasis*, *TrailingTerm*, *UnivariatePolynomial*, *Walk*, *WeightedDegree*]

The commands that we will mainly use are
Basis - for computing a Groebner basis
NormalForm - for computing the remainder of a polynomial divided by a (Groebner) basis

Lets execute Buchberger's algorithm on the ideal I = <f_1, f_2> below.
.

```
> f1 := x*y-y^2;
  f2 := x^3-z^2;
  G0 := [f1,f2];
```

$$f1 := x\,y - y^2$$

$$f2 := x^3 - z^2$$

$$G0 := [x\,y - y^2,\ x^3 - z^2]$$

We'll use the SPolynomial and LeadingMonomial commands.
And we'll use lexicographical order with $x > y > z$

```
> LeadingMonomial(f1,plex(x,y,z));
  LeadingMonomial(f2,plex(x,y,z));
```

$$x\,y$$

$$x^3$$

```
> f3 := SPolynomial(f1,f2,plex(x,y,z));
```

$$f3 := -x^2\,y^2 + y\,z^2$$

The NormalForm computes the remainder of S(f1,f2) divided by G.

```
> f3 := NormalForm(f3,G0,plex(x,y,z));
```

$$f3 := -y^4 + y\,z^2$$

The remainder is not 0 so we add f3 to the basis.

```
> G1 := [f1,f2,f3];
```

$$G1 := [x\,y - y^2,\ x^3 - z^2,\ -y^4 + y\,z^2]$$

```
> map(LeadingMonomial,G1,plex(x,y,z));
```

$$[x\,y,\ x^3,\ y^4]$$

Now S(f2,f3) reduces to 0 by Proposition 4 of 2.9 so we only need to consider

```
> f4 := SPolynomial(f1,f3,plex(x,y,z));
```

$$f4 := y^5 - x\,y\,z^2$$

```
> f4 := NormalForm(f4,G1,plex(x,y,z));
```

$$f4 := 0$$

So G1 is a Groebner basis for I = <f1,f2>. It happens to be also reduced. Let's check with Maple.

```
> G := Basis([f1,f2],plex(x,y,z));
```

$$G := [y^4 - y\,z^2,\ x\,y - y^2,\ x^3 - z^2]$$

```
> G1;
```

$$[x\,y - y^2,\ x^3 - z^2,\ -y^4 + y\,z^2]$$

This polynomial is in the ideal I. Let's test this

```
> f := expand(x*f1+y*f2);
```

$$f := x^3\,y + x^2\,y - x\,y^2 - y\,z^2$$

```
> NormalForm(f,G,plex(x,y,z));
```

$$0$$

```
> NormalForm(f+x+1,G,plex(x,y,z));
```

$$1 + x$$

Monomial Orderings in Maple

$k[x, y, z]$	Cox, Little O'Shea text	Maple
Lexicographical order	lex with $x > y > z$	plex(x,y,z)
Graded lexicographical order	grlex with $x > y > z$	grlex(x,y, z)
Graded reverse lexicographical order	grevlex with $x > y > z$	tdeg(x,y,z)

```
> f := 3*x^3 + 4*x*y*z^2 + 5*y^3*z;
```

$$f := 4\,x\,y\,z^2 + 5\,y^3\,z + 3\,x^3$$

Notice that the LeadingTerm command in Maple returns a pair (leading coefficient,

```
> LeadingMonomial(f,plex(x,y,z));
  LeadingCoefficient(f,plex(x,y,z));
  LeadingTerm(f,plex(x,y,z));
```

$$x^3$$

$$3$$

$$3,\ x^3$$

```
> LeadingMonomial(f,grlex(x,y,z));
  LeadingCoefficient(f,grlex(x,y,z));
```

$$x\,y\,z^2$$

$$4$$

```
> LeadingMonomial(f,tdeg(x,y,z));
  LeadingCoefficient(f,tdeg(x,y,z));
```

$$y^3 z$$

$$5$$

Graph Coloring

Let's try to color the graph C3, a cycle on three vertices with k=2 colors (it's not 2-colorable).




```
> k := 2;
  S := [x1^k-1,x2^2-1,x3^2-1];
```

$$k := 2$$

$$S := [x1^2 - 1, x2^2 - 1, x3^2 - 1]$$

The colors are the roots of $x^2 - 1$ which are 1 and -1. These equations say each vertex can be colored with either color. But vertex 1 may not be the same color as vertex 2. So we should add the equation x1 + x2 = 0. Same for the other two edges. We have

```
> S := [op(S), x1+x2, x1+x3, x2+x3 ];
```

$$S := [x1^2 - 1, x2^2 - 1, x3^2 - 1, x1 + x2, x1 + x3, x2 + x3]$$

```
> Basis(S,grlex(x1,x2,x3));
```

$$[1]$$

This means G is not 2-colorable. But it is 3-colorable.

```
> k := 3;
  S := [x1^k-1,x2^k-1,x3^k-1,
        normal((x1^k-x2^k)/(x1-x2)),
        normal((x1^k-x3^k)/(x1-x3)),
        normal((x2^k-x3^k)/(x2-x3))];
```

$$k := 3$$

$$S := [x1^3 - 1, x2^3 - 1, x3^3 - 1, x1^2 + x2\,x1 + x2^2, x1^2 + x1\,x3 + x3^2, x3^2 + x2\,x3 + x2^2]$$

```
> Basis(S,grlex(x1,x2,x3));
```

$$[x1 + x3 + x2, x3^2 + x2\,x3 + x2^2, x3^3 - 1]$$

This means x3 can be any color, x2 must be different from x3, and x1, x2, x3 must have all different colors.

Solving Polynomial Equations using lex Groebner bases.

Lets work with one of the ideals we saw in class, namely

I = < $x^2+y+z-1,\ x+y^2+z-1,\ x+y+z^2-1$ >

```
> F := [x^2+y+z-1,x+y^2+z-1,x+y+z^2-1];
```

$$F := [x^2+y+z-1,\ y^2+x+z-1,\ z^2+x+y-1]$$

```
> G := Basis(F,grlex(x,y,z));
```

$$G := [z^2+x+y-1,\ y^2+x+z-1,\ x^2+y+z-1]$$

```
> H := Basis(F,plex(x,y,z));
```

$$H := [z^6-4z^4+4z^3-z^2,\ z^4+2yz^2-z^2,\ y^2-z^2-y+z,\ z^2+x+y-1]$$

We can see that the first polynomial has repeated roots.

```
> factor(H[1]);
```

$$z^2(z^2+2z-1)(z-1)^2$$

The Solve command in the Groebner basis package drops multiple solutions.

```
> Solve(H,[x,y,z]);
```

$$\{[[z, y, -1+x], plex(x,y,z), \{\ \}], [[z, -1+y, x], plex(x,y,z), \{\ \}], [[z-1, y, x], plex(x,y,z), \{\ \}], [[z^2+2z-1, y-z, -z+x], plex(x,y,z), \{z, z-1\}]\}$$

This has split up the solutions. We can solve them explicitly using solve

```
> _EnvExplicit := true;
  V := {solve(H,{x,y,z})};
```

$$\_EnvExplicit := true$$

$$V := \{\{x=0, y=0, z=1\}, \{x=0, y=1, z=0\}, \{x=1, y=0, z=0\}, \{x=-1-\sqrt{2}, y=-1-\sqrt{2}, z=-1-\sqrt{2}\}, \{x=\sqrt{2}-1, y=\sqrt{2}-1, z=\sqrt{2}-1\}\}$$

```
> nops(V);
```

5

There are 5 distinct solutions in the variety V.

The PolynomialIdeals package.

The PolynomialIdeals package has additional operations for ideals.
It also allows me to use < > brackets for ideals.

```
> with(PolynomialIdeals);
```

[<,>, *Add*, *Contract*, *EliminationIdeal*, *EquidimensionalDecomposition*, *Generators*, *HilbertDimension*, *IdealContainment*, *IdealInfo*, *IdealMembership*, *Intersect*, *IsMaximal*, *IsPrimary*, *IsPrime*, *IsProper*, *IsRadical*, *IsZeroDimensional*, *MaximalIndependentSet*, *Multiply*, *NumberOfSolutions*, *Operators*, *PolynomialIdeal*, *PrimaryDecomposition*, *PrimeDecomposition*, *Quotient*, *Radical*, *RadicalMembership*, *Saturate*, *Simplify*, *UnivariatePolynomial*, *VanishingIdeal*, *ZeroDimensionalDecomposition*, *in*, *subset*]

```
> interface(imaginaryunit = iii);
```

iii

The above is to let me use the capital letter I for an ideal.

```
> F;
```

$$[x^2+y+z-1,\ y^2+x+z-1,\ z^2+x+y-1]$$

```
> I := <F>;
```
$$I := \langle z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle$$

The radical operation gets rid of all repeated solutions.

```
> J := Radical(I);
```
$$J := \langle z^4 + z^3 - 3\,z^2 + z,\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle$$

Compute a Groebner basis for J

```
> Basis(J,plex(x,y,z));
```
$$[z^4 + z^3 - 3\,z^2 + z,\ z^3 + 2\,y\,z - z,\ y^2 - z^2 - y + z,\ z^2 + x + y - 1]$$

The PrimeDecompostion operation splits the ideal J into prime components, each of which corresponds to an irreducible variety.

```
> P := [PrimeDecomposition(J)];
```
$$P := [\langle z - 1,\ z^4 + z^3 - 3\,z^2 + z,\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle,\ \langle z^2 + 2\,z - 1,\ z^4 + z^3 - 3\,z^2 + z,\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle,\ \langle y,\ z,\ z^4 + z^3 - 3\,z^2 + z,\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle,\ \langle z,\ -1 + y,\ z^4 + z^3 - 3\,z^2 + z,\ z^2 + x + y - 1,\ y^2 + x + z - 1,\ x^2 + y + z - 1\rangle]$$

```
> P := map(Simplify,P);
```
$$P := [\langle x,\ y,\ z - 1\rangle,\ \langle y - z,\ -z + x,\ z^2 + 2\,z - 1\rangle,\ \langle y,\ z,\ -1 + x\rangle,\ \langle x,\ z,\ -1 + y\rangle]$$

From which we can see 1, 1, 2, 1 solutions. From which we can understand the output of the Solve command above.